

**Nombrar Rangos de Celdas** sirve para simplificar las funciones de búsqueda en casos en los que se utilizará la misma matriz de datos en varias fórmulas

*Por ejemplo, si nombramos el rango de A1:D6 “Ropa”...*

	Ropa	X	✓	fx	Producto
	A	B	C	D	
1	Producto	Cantidad	SKU	Precio	
2	Camisa	26	34564	\$14.99	
3	Pantalón	14	23534	\$49.99	
4	Shorts	22	84563	\$24.50	
5	Calcetas	36	25768	\$9.99	
6	Gorra	2	34684	\$29.99	

*...se puede escribir la función de BUSCARV de las siguientes dos maneras:*

**=BUSCARV(A1,\$A\$1:\$D\$6,2)**

**=BUSCARV (A1,Ropa,2)**

Exploremos una de las funciones de referencia más comunes de Excel – **BUSCARV**:

**=BUSCARV(valor\_buscado, matriz\_tabla, indicador\_columnas, [rango])**

El **valor** que se buscará en la matriz seleccionada

Aquí es **donde** se buscará el valor

¿**Qué columna** tiene los datos que buscas extraer?

¿Quieres que busque **exactamente** el mismo valor (0), o algo similar (1)?

	A	B	C	D
1	Producto	Cantidad	SKU	Precio
2	Camisa	26	34564	\$14.99
3	Pantalón	14	23534	\$49.99
4	Shorts	22	84563	\$24.50
5	Calcetas	36	25768	\$9.99
6	Gorra	2	34684	\$29.99

**=BUSCARV(A2, \$G\$1:\$H\$5, 2, 0)**

G	H
Producto	Precio
Shorts	\$24.50
Pantalón	\$49.99
Gorra	\$29.99
Camisa	\$14.99
Calcetas	\$9.99

*Para poblar la columna D con los precios, se busca el nombre del producto en el rango G1:H5 y se devuelve el valor de la 2da columna de izquierda a derecha*

Usa **BUSCARH** si tu matriz está transpuesta (los títulos escritos como filas)

**=BUSCARH(valor\_buscado, matriz\_buscar\_en, indicador\_filas, [ordenado])**

El **valor** que se buscará en la matriz seleccionada

Aquí es **donde** se buscará el valor

¿**Qué fila** tiene los datos que buscas extraer?

¿Quieres que busque **exactamente** el mismo valor (0), o algo similar (1)?

	A	B	C	D
1	Producto	Cantidad	SKU	Precio
2	Camisa	26	34564	\$14.99
3	Pantalón	14	23534	\$49.99
4	Shorts	22	84563	\$24.50
5	Calcetas	36	25768	\$9.99
6	Gorra	2	34684	\$29.99

**=BUSCARH(A2, \$H\$1:\$L\$2, 2, 0)**

*Al usar BUSCARH, se busca el producto en el rango H1:L2 y se devuelve el valor de la 2da fila de arriba para abajo*

G	H	I	J	K	L
Producto	Shorts	Pantalón	Gorra	Camisa	Calcetas
Precio	\$24.50	\$49.99	\$29.99	\$14.99	\$9.99

Hay **dos reglas clave** que limitan las funciones de **BUSCARH** y **BUSCARV**:



1. El valor de búsqueda debe estar en la **primer columna** de la matriz de datos para BUSCARV o en la **primera fila** de la matriz para BUSCARH
2. Excel siempre devolverá el valor en la **fila más alta** o en la **columna más a la izquierda** de la matriz cuando hay varias coincidencias con el valor de búsqueda



### TIP EXPERTO:

*Evita violar la segunda regla identificando una “Llave” que compartan ambas matrices y sea única por fila (NOTA: Las llaves muchas veces son campos concatenados de cada matriz)*

La función **FILA** devuelve el número de fila dada una *referencia*, mientras que **FILAS** devuelve el número de filas en una *matriz* o *formula matricial*

=FILA([ref])

=FILAS(matriz)

*Este ejemplo utiliza una matriz, por lo que incluye los símbolos de { } – se verá en la sección de Fórmulas Matriciales*

FILA(C10) = 10

FILAS(A10:D15) = 6

FILAS({1,2,3;4,5,6}) = 2



La función **COLUMNA** devuelve el número de columna dada una *referencia*, mientras que **COLUMNAS** devuelve el número de columnas en una *matriz* o *formula matricial*

=COLUMNA([ref])

=COLUMNAS(matriz)



### TIP EXPERTO:

Usa **FILA()** o **COLUMNA()** sin referencia para devolver el número de fila o columna de la celda en la que está escrita la fórmula

COLUMNA(C10) = 3

COLUMNAS(A10:D15) = 4

COLUMNAS({1,2,3;4,5,6}) = 3

La función **INDICE** devuelve el *valor* de una celda específica en una matriz

**=INDICE(matriz, num\_fila, num\_columna)**

¿En qué rango de celdas nos estamos enfocando?

¿En qué fila está el valor que queremos?

¿En qué columna está el valor que queremos?

	A	B	C
1	Herramienta	Precio	Inventario
2	Martillo	\$5.00	55
3	Taladro	\$10.00	66
4	Pinza	\$2.50	333
5	Llave	\$6.00	234
6	Nivel	\$12.00	23
7	Sierra	\$16.00	5

**INDICE(\$A\$1:\$C\$5, 5, 3) = 234**

*En este caso le estamos diciendo a Excel que encuentre el valor de una celda en el rango A1:C5. Empezando desde la esquina superior izquierda bajamos a la **5<sup>ta</sup> fila** y nos desplazamos a la derecha a la **3<sup>er</sup> columna**, devolviendo el valor de **234***

La función **COINCIDIR** devuelve la *posición* de un valor específico dentro de una fila o columna

**=COINCIDIR(valor\_buscado, matriz\_buscada, [tipo\_coincidencia])**

¿De qué valor buscas la posición?

¿En qué fila o columna estás buscando? (**debe ser matriz unidimensional**)

¿Estás buscando el valor exacto (0), o algo similar?

1: Busca el valor más grande < o = valor especificado

0: Busca el valor especificado

-1: Busca el valor más chico > o = valor especificado

	A	B
1	Herramienta	Precio
2	Martillo	\$5.00
3	Taladro	\$10.00
4	Pinza	\$2.50
5	Llave	\$6.00

**COINCIDIR("Pinza", \$A\$1:\$A\$5, 0) = 4**

	A	B	C
1	Herramienta	Precio	Inventario
2	Martillo	\$5.00	55
3	Taladro	\$10.00	66
4	Pinza	\$2.50	333

**COINCIDIR(66, \$A\$3:\$C\$3, 0) = 3**

Al buscar "Pinza" en la columna A, la encontramos en la **4<sup>ta</sup> fila**. Al buscar el número 66 en la fila 3, lo encontramos en la **3<sup>er</sup> columna**



**INDICE** y **COINCIDIR** se pueden usar en conjunto como una función BUSCAR, solo que **INDICE/COINCIDIR** puede buscar valores en cualquier fila o columna

*Ejemplo: Cotizador*

	A	B	C	D
1		S	M	L
2	Camisa	\$10	\$12	\$15
3	Calcetas	\$30	\$35	\$40
4	Pants	\$25	\$30	\$35
5				
6	Producto:	Pants		
8	Talla:	M		
10	PRECIO:	?		

*En el ejemplo queremos obtener el precio de un producto con tamaño determinado en la celda B10 al buscar en el rango B2:D4*

**B10=INDICE(B2:D4, COINCIDIR(B6,A2:A4,0), COINCIDIR(B8,B1:D1,0))**

*Como el número de la fila depende del producto, usaremos COINCIDIR para buscar el valor de la celda B6 (en este caso "Pants")*

*El número de columna depende de la talla, así que se usa COINCIDIR para buscar el valor de la celda B8 (en este caso, "M")*

*Al considerar el resultado de cada función COINCIDIR, la fórmula se resume en un INDICE:*

**B10 = INDICE(B2:D4, 3, 2) = \$30**

La función **BUSCARX** devuelve valores de un rango al coincidir un valor de búsqueda, y ofrece mayor flexibilidad que las fórmulas de **BUSCARV**, **BUSCARH**, **INDICE** y **COINCIDIR**

**=BUSCARX(valor\_buscado, matriz\_buscada, matriz\_devuelta, [si\_no\_se\_encuentra], [modo\_de\_coincidencia], [modo\_de\_búsqueda])**

¿Qué **valor** buscas que encuentre?

¿**Dónde** estas tratando de encontrar ese valor?

¿Dónde están los **valores** que quieres devolver?

¿Y si **no se encuentra** el valor buscado en la matriz de búsqueda?

¿Estás buscando una coincidencia **exacta**, **aproximada**, o de **comodín**?

¿Quieres que busque de **arriba hacia abajo** o **abajo hacia arriba**?

**IMPORTANTE:** BUSCARX solo está disponible para suscriptores de **Microsoft 365**

## BUSCARX

- ✓ Puede devolver una **matriz dinámica** de resultados
- ✓ Puede buscar valores **en cualquier parte** de una matriz (izquierda o derecha, horizontal o vertical)
- ✓ **Coincidencia exacta** predeterminada
- ✓ Permite coincidencias de texto con **comodines**
- ✓ Incluye **manejo de errores** cuando no se encuentra el valor buscado
- ✓ Encuentra coincidencias aproximadas en listas **desordenadas**
- ✓ Puede buscar de forma **descendente** o **ascendente**

## BUSCARV

- ✗ Devuelve **un solo valor**
- ✗ Solo puede buscar valores a la **derecha**, requiere **BUSCARH** para coincidencias horizontales
- ✗ **Coincidencia aproximada** predeterminada
- ✗ No permite coincidencias de texto con **comodines**
- ✗ Requiere una función de **SIERROR** para el manejo de errores
- ✗ Necesita listas **ordenadas** para coincidencias aproximadas
- ✗ Solo busca de manera **descendente**

La función **ELEGIR** selecciona un valor, una referencia o una formula para ejecutar en base a un número de índice dado

**=ELEGIR(núm\_índice, valor1, [valor2], ...)**

¿Cuál elemento de la siguiente lista debe ser evaluado?

1<sup>er</sup> elemento en la lista

2<sup>do</sup> elemento en la lista

3<sup>ro</sup>, 4<sup>to</sup>, 5<sup>to</sup>, etc...

### DATOS INTERESANTES DE ELEGIR:

- Los elementos de la lista pueden incluir **números, referencias, rangos nombrados, fórmulas, o texto** (o una mezcla!)
- ELEGIR actúa como la función de **INDIRECTO**, y puede interpretar referencias a celdas en lugar de verlas como texto
- Puedes combinar ELEGIR con otras funciones, o anidarla directamente a una referencia

La función **DESREF** es similar a **INDICE**, pero puede devolver tanto el valor de una celda (cómo **INDICE**) como un *rango* específico de celdas

**=DESREF(ref, filas, columnas, [alto], [ancho])**

¿Dónde  
comienzas?

¿Cuántas filas te  
quieres desplazar?

¿Cuántas columnas te  
quieres desplazar?

¿Si quieres devolver un rango,  
que tan alto y ancho debe ser?

Una función **DESREF** donde **[alto]=1** y **[ancho]=1** funciona exactamente igual a **INDICE**. Un uso común de **DESREF** es para crear matrices dinámicas (cómo el ejemplo de la Tabla Navegable)



### TIP EXPERTO:

*No uses DESREF o INDICE/COINCIDIR cuando se pueda usar BUSCARV*