

Las funciones de **ESPACIOS**, **MAYUSC**, **MINUSC**, y **NOMPROPIO** son utilizadas para estandarizar el formato de las celdas de texto

	A	B	C	D
1	Texto de ejemplo	Formula	Resultado	Notas
2	Enunciado de MUESTRA	=ESPACIOS(A2)	Enunciado de MUESTRA	Quita todos los espacios del texto excepto los espacios individuales entre palabras
3	Enunciado de MUESTRA	=MINUSC(A3)	enunciado de muestra	Convierte todas la letras de una cadena de texto en minúsculas
4	Enunciado de MUESTRA	=MAYUSC(A4)	ENUNCIADO DE MUESTRA	Convierte todas la letras de una cadena de texto en mayúsculas
5	Enunciado de MUESTRA	=NOMPROPIO(A5)	Enunciado De Muestra	Convierte la primer letra de cada palabra en mayúscula y el resto en minúsculas



## TIP EXPERTO:

*Si dos cadenas de texto son idénticas con la excepción de un espacio al final de una, se verán exactamente iguales pero Excel las tratara como diferentes; ESPACIOS hará que sean iguales*

**CONCATENAR** te permite combinar texto, valores de celdas, o resultados de fórmulas en una sola cadena de texto

**Nota:** En vez de usar "**=CONCATENAR**(*texto1, texto2...*)", puedes unir cada uno de los textos con el símbolo "&".

	A	B	C	D
1	Nombre	Apellido	Formula	Resultado
2	Daniel	Ramos	=A2&B2	DanielRamos
3	Daniel	Ramos	=A3&" "&B3	Daniel Ramos
4	Daniel	Ramos	=IZQUIERDA(A4,3)&" "&B4	Dan Ramos
5	Daniel	Ramos	=IZQUIERDA(A5,3)&" "&IZQUIERDA(B5,1)&"."	Dan R.

Las funciones de **IZQUIERDA**, **DERECHA**, y **EXTRAER** devuelven un número específico de caracteres a partir de una ubicación en una cadena de texto, mientras que **LARGO** devuelve el número de caracteres

**=IZQUIERDA(texto, [num\_de\_caracteres])**

**=EXTRAER(texto, posición\_inicial, num\_de\_caracteres)**

**=LARGO(texto)**

	A	B	C	D
1	Texto de ejemplo	Formula	Resultado	Notas
2	MA-02215%AAA%_100	=IZQUIERDA(A2,2)	MA	Devuelve dos caracteres, empezando desde la izquierda
3	MA-02215%AAA%_100	=EXTRAER(A3,4,5)	02215	Devuelve cinco caracteres, empezando desde el cuarto
4	MA-02215%AAA%_100	=DERECHA(A4,3)	100	Devuelve tres caracteres, empezando desde la derecha
5	MA-02215%AAA%_100	=LARGO(A5)	17	Devuelve el número de caracteres en la cadena de texto

La función de **TEXTO** convierte un valor numérico a texto y le asigna un formato

**=TEXTO(valor, formato)**

Valor numérico, formula que resulta en un número, o referencia a una celda con valor numérico

Formato numérico expresado como texto envuelto en comillas (ej. "d/m/yyyy", "\$0.00" o "#,##0.00")

	A	B
1	Nombre	Ganancias
2	Luis	\$4,500
3	Carlos	\$3,250
4	José	\$3,725

= "José gana "&B4 *devuelve* "José gana 3725"

= "José gana "&**TEXTO(B4"\$#,###")** *devuelve* "José gana \$3,725"



### TIP EXPERTO:

Usa **VALOR** para convertir una cadena de texto que represente un número a un valor

La función de **HALLAR** devuelve la posición en la que se encuentra un carácter o una cadena de texto (si no lo encuentra devuelve un error de #¡VALOR!)

**=HALLAR(texto\_buscado, dentro\_del\_texto, [num\_inicial])**

¿Qué carácter o cadena de texto buscas?

¿Dónde está el texto en el que buscarás?

¿Buscar desde el principio (default) o después de un cierto número de caracteres?

	A	B	C	D
1	Texto de ejemplo	Formula	Resultado	Notas
2	MA-02215%AAA%_100	=HALLAR("%",A2)	9	Busca la cadena de texto "%" y devuelve su posición
3	MA-02215%AAA%_100	=HALLAR("%",A3,10)	13	Busca "%", empezando desde el 10mo caracter, y devuelve su posición
4	MA-02215%AAA%_100	=EXTRAER(A4,HALLAR("%",A4),5)	%AAA%	Devuelve 5 caracteres del centro del texto, empezando del "%"
5	MA-02215%AAA%_100	=EXTRAER(A5,HALLAR("%",A5)+1,3)	AAA	Devuelve 3 caracteres del centro del texto, empezando después del "%"



### TIP EXPERTO:

**ENCONTRAR** funciona exactamente igual, pero considera mayúsculas

**SI(ESNUMERO(HALLAR(** es una combinación de funciones que se puede usar para clasificar datos basados en celdas que contienen cadenas específicas de texto

**=SI(ESNUMERO(HALLAR(texto\_buscado, dentro\_del\_texto)),valor\_si\_verdadero, valor\_si\_falso)**

Busca una cadena de texto en una celda

Devuelve un valor si encuentra la cadena de texto (VERDADERO), y otro si no la encuentra (FALSO)

	A	B
1	Código	Tipo
2	13816-Aff-160x90_chica	Otro
3	6121-Disp-160x90_grande	Display
4	984131-Aff-160x90_chica	Otro
5	16812-Disp-160x90_chica	Display
6	98542-Agg-160x90_grande	Otro

**=SI(ESNUMERO(HALLAR("Disp",A2)),"Display","Otro")**

*Busca la cadena de texto "Disp" en las celdas de la columna A y clasifica la columna B como "Display" si la encuentra, y como "Otro" si no la encuentra*