

Las funciones de **CONTAR**, **PROMEDIO**, **MEDIANA**, **MODA**, **MAX/MIN**, **PERCENTIL** y **DESVEST/VAR** son usadas para hacer cálculos básicos en una matriz de datos

	A	B	C	D
1	Valor			
2	90	Tamaño de Muestra:	19	=CONTAR(A2:A20)
3	13			
4	22	Promedio:	51.47	=PROMEDIO(A2:A20)
5	98			
6	61	Mediana:	54	=MEDIANA(A2:A20)
7	68			
8	50	Moda:	22	=MODA(A2:A20)
9	91			
10	16	Max:	98	=MAX(A2:A20)
11	23			
12	60	Min:	13	=MIN(A2:A20)
13	22			
14	56	25 Percentil:	23	=PERCENTIL(A2:A20,.25)
15	54			
16	87	75 Percentil:	68	=PERCENTIL(A2:A20,.75)
17	33			
18	68	Desviación Estándar:	28	=DESVEST(A2:A20)
19	45			
20	21	Varianza:	767	=VAR(A2:A20)

	A
1	Valor
2	90
3	13
4	22
5	98
6	61
7	68
8	50

→ JERARQUIA(A2,A2:A8) = 2
 → JERARQUIA(A3,A2:A8) = 7 (menor)
 → JERARQUIA(A4,A2:A8) = 6
 → JERARQUIA(A5,A2:A8) = 1 (mayor)
 → JERARQUIA(A6,A2:A8) = 4
 → JERARQUIA(A7,A2:A8) = 3
 → JERARQUIA(A8,A2:A8) = 5

La función **JERARQUIA** devuelve la jerarquía de un número dentro de una lista de valores

Las funciones **K.ESIMO.MAYOR** y **K.ESIMO.MENOR** devuelven el k-ésimo valor más chico/grande de una matriz

	A
1	Valor
2	90
3	13
4	22
5	98
6	61
7	68
8	50

K.ESIMO.MAYOR(A2:A8,2) = 90
 (el 2do número más grande de la matriz es 90)

K.ESIMO.MENOR(A2:A8,3) = 50
 (el 3er número más chico de la matriz es 50)

	A	B
1	Valor	Rango Percentil
2	2,717	18%
3	3,485	24%
4	5,202	76%
5	3,612	29%
6	4,432	59%
7	2,699	12%
8	4,585	65%
9	6,003	94%
10	4,820	71%
11	2,550	6%
12	5,795	88%
13	4,240	41%
14	6,827	100%
15	4,359	53%
16	2,320	0%
17	5,775	82%
18	4,241	47%
19	3,966	35%

RANGO.PERCENTIL devuelve la jerarquía de un número dentro de un conjunto de datos como porcentaje del conjunto

=RANGO.PERCENTIL(matriz, x)

¿Qué rango de datos estás usando?

¿Qué **valor** dentro del rango quieres evaluar?

RANGO.PERCENTIL(\$A\$2:\$A\$19, A14) = 100% (mayor)

RANGO.PERCENTIL(\$A\$2:\$A\$19, A16) = 0% (menor)

ALEATORIO y **ALEATORIO.ENTRE** son las funciones generadoras de números aleatorios de Excel:

	A	B	C	D	E
1	0.5173	0.4091	0.7560	0.9012	0.2167
2	0.0906	0.2317	0.0906	0.5856	0.8646
3	0.1544	0.8240	0.4279	0.8782	0.7795
4	0.0097	0.0872	0.7740	0.9137	0.7815
5	0.2089	0.7028	0.0449	0.8173	0.9983
6	0.0761	0.4388	0.4056	0.5639	0.0668

La función **ALEATORIO()** devuelve un valor aleatorio entre 0 and 1 (de 15 dígitos)

La función de **ALEATORIO.ENTRE** devuelve un número entero entre dos valores especificados

	A	B	C	D	E
1	83	23	64	62	92
2	59	45	40	50	91
3	24	37	70	30	32
4	54	85	69	55	3
5	73	12	36	53	2
6	29	72	68	59	99

=ALEATORIO.ENTRE(0,100)

La función de **SUMAPRODUCTO** multiplica las celdas correspondientes entre múltiples matrices y devuelve la suma de los productos (*Nota: las matrices deben de tener las mismas dimensiones*)

=SUMAPRODUCTO(matriz1, matriz2, ... matriz_n)

Ejemplo: Ingreso Total

	A	B	C	D
1	Producto	Cantidad	Precio	Ingreso
2	A	2	\$7.00	\$14.00
3	B	4	\$3.00	\$12.00
4	C	3	\$12.50	\$37.50
5	Total			\$63.50

Sin usar SUMAPRODUCTO, se puede multiplicar la cantidad por el precio en cada fila y al final sumar los ingresos

	A	B	C	D
1	Producto	Cantidad	Precio	Ingreso
2	A	2	\$7.00	
3	B	4	\$3.00	
4	C	3	\$12.50	
5	Total			\$63.50

SUMAPRODUCTO(B2:B4,C2:C4) = \$63.50

SUMAPRODUCTO se usa comúnmente con filtros para calcular productos *únicamente* de las filas que cumplan ciertos criterios:

	A	B	C	D
1	Tienda	Producto	Cantidad	Precio
2	Super Fresco	Manzana	2	\$10.00
3	Mercadito	Fresas	4	\$20.00
4	La Esquina	Fresas	3	\$20.00
5	Frutas del Juan	Piña	8	\$25.00
6	Super Fresco	Naranja	2	\$15.00
7	Mercadito	Manzana	1	\$10.00
8	La Esquina	Manzana	5	\$10.00
9	Frutas del Juan	Fresas	6	\$20.00
10	La Esquina	Piña	3	\$25.00
11	Frutas del Juan	Naranja	8	\$15.00
12	Super Fresco	Piña	3	\$25.00
13	Mercadito	Piña	5	\$25.00
14	Super Fresco	Fresas	2	\$20.00
15	Mercadito	Naranja	6	\$15.00
16	La Esquina	Naranja	7	\$15.00
17	Frutas del Juan	Manzana	3	\$10.00

Cantidad de productos vendidos en Mercadito:

SUMAPRODUCTO((A2:A17="Mercadito")*C2:C17) = 16

Ingresos totales de Mercadito:

SUMAPRODUCTO((A2:A17="Mercadito")*C2:C17*D2:D17) = \$305

Ingresos totales de manzanas vendidas en Mercadito:

SUMAPRODUCTO((A2:A17="Mercadito")*(B2:B17="Manzana")*C2:C17*D2:D17) = \$10

Sobre todo este rango de valores, para cada celda asigna 1 si es True, 0 si es False
Luego procede a multiplicar ese valor por las demás restricciones (que están separadas con *)



TIP EXPERTO:

Al agregar filtros a SUMAPRODUCTO, se necesitan cambiar las comas a signos de multiplicación

Excelente pero, ¿cómo funciona?

SUMAPRODUCTO((A2:A17="Mercadito")*(B2:B17="Manzana")*C2:C17*D2:D17) = \$10

	A	B	C	D
1	Tienda	Producto	Cantidad	Precio
2	Super Fresco	Manzana	2	\$10.00
3	Mercadito	Fresas	4	\$20.00
4	La Esquina	Fresas	3	\$20.00
5	Frutas del Juan	Piña	8	\$25.00
6	Super Fresco	Naranja	2	\$15.00
7	Mercadito	Manzana	1	\$10.00
8	La Esquina	Manzana	5	\$10.00
9	Frutas del Juan	Fresas	6	\$20.00
10	La Esquina	Piña	3	\$25.00
11	Frutas del Juan	Naranja	8	\$15.00
12	Super Fresco	Piña	3	\$25.00
13	Mercadito	Piña	5	\$25.00
14	Super Fresco	Fresas	2	\$20.00
15	Mercadito	Naranja	6	\$15.00
16	La Esquina	Naranja	7	\$15.00
17	Frutas del Juan	Manzana	3	\$10.00

Lo que TÚ ves

Lo que EXCEL ve

Al aplicarle un filtro o condición a una columna, Excel convierte las celdas que son falsas a 0 y las verdaderas a 1.

Si multiplicas las 4 columnas, SÓLO LAS FILAS EN LAS QUE SE CUMPLAN TODAS LAS CONDICIONES DARÁN UNA SUMA QUE NO SEA CERO

	A	B	C	D
1	Tienda	Producto	Cantidad	Precio
2	0	1	2	\$10.00
3	1	0	4	\$20.00
4	0	0	3	\$20.00
5	0	0	8	\$25.00
6	0	0	2	\$15.00
7	1	1	1	\$10.00
8	0	1	5	\$10.00
9	0	0	6	\$20.00
10	0	0	3	\$25.00
11	0	0	8	\$15.00
12	0	0	3	\$25.00
13	1	0	5	\$25.00
14	0	0	2	\$20.00
15	1	0	6	\$15.00
16	0	0	7	\$15.00
17	0	1	3	\$10.00

Las funciones de **CONTAR.SI**, **SUMAR.SI**, y **PROMEDIO.SI** calculan la cuenta, la suma o el promedio basadas en criterios específicos.

	A	B
1	Nombre	Edad
2	Jorge	90
3	Maria	13
4	Luis	22
5	Carlos	98
6	Jorge	61
7	Luis	68
8	Luis	50
9	Maria	91
10	Jorge	16
11	Maria	23
12	Luis	60
13	Carlos	22
14	Maria	56
15	Jorge	54
16	Jorge	87
17	Carlos	33
18	Carlos	68
19	Carlos	45
20	Jorge	21

=CONTAR.SI(rango, criterio)

=SUMAR.SI(rango, criterio, rango_suma)

=PROMEDIO.SI(rango, criterio, rango_promedio)

¿Qué celdas deben cumplir con el criterio?

¿Bajo qué criterio quiero que se haga el cálculo?

¿Dónde están los valores con los que haré el cálculo?

CONTAR.SI(B2:B20,22) = 2

SUMAR.SI(A2:A20,"Luis",B2:B20) = 190

SUMAR.SI(A2:A20,"<>Carlos",B2:B20) = 702

PROMEDIO.SI(A2:A20,"Maria",B2:B20) = 45.75

CONTAR.SI.CONJUNTO, SUMAR.SI.CONJUNTO, y PROMEDIO.SI.CONJUNTO se usan para evaluar las funciones utilizando *múltiples* criterios.

=CONTAR.SI.CONJUNTO(rango_criterios1, criterio1, rango_criterios2 , criterio2...)

=SUMAR.SI.CONJUNTO(rango_suma, rango_criterios1, criterio1, rango_criterios2 , criterio2...)

=PROMEDIO.SI.CONJUNTO(rango_promedio, rango_criterios1, criterio1, rango_criterios2 , criterio2...)

	A	B	C	D
1	Mes	Duración	Red Social	Likes
2	Ene	Larga	Facebook	166
3	Ene	Larga	Youtube	263
4	Ene	Corta	Instagram	289
5	Ene	Corta	Snapchat	137
6	Feb	Larga	Facebook	124
7	Feb	Larga	Youtube	311
8	Feb	Corta	Instagram	350
9	Feb	Corta	Snapchat	384
10	Mar	Larga	Facebook	168
11	Mar	Larga	Youtube	358
12	Mar	Corta	Instagram	347
13	Mar	Corta	Snapchat	390

Se deben cumplir **AMBAS** condiciones para ser tomadas en cuenta

CONTAR.SI.CONJUNTO(B2:B13, "Larga", D2:D13, ">200") = 3

SUMAR.SI.CONJUNTO(D2:D13, A2:A13, "Feb", B2:B13, "Corta") = 734

PROMEDIO.SI.CONJUNTO(D2:D13, A2:A13, "Ene", C2:C13, "Youtube") = 263



TIP EXPERTO:

Para usar < o > se necesitan usar comillas, como si fuera texto (ej. ">200") *Obs.*